
Panache thermique

1 Mouvements convectifs dans un liquide

1.1 Introduction

Le montage expérimental proposé illustre de manière simple la mise en mouvement d’un fluide
par convection (Fig. 1a). Ce fluide (ici de l’eau) est contenu dans une enceinte ponctuellement
chauffée dans sa partie inférieure (la surface supérieure du liquide est en contact avec l’air
ambient de température inférieure). La densité du liquide diminue légèrement au voisinage du
point chaud, ce qui engendre son ascension à la manière d’une montgolfière. Ce phénomène
est omniprésent dans la vie quotidienne : il fait monter la fumée à la sortie d’une cheminée
et conduit au dépôt de poussières au dessus des radiateurs. À bien plus grande échelle, ces
mouvements convectifs sont à l’origine, par exemple, de la brise thermique en bord de mer,
des mouvements internes au manteau terrestre (Fig. 1b).

Ce montage permet également de mettre en œuvre une technique très utilisée en mécanique
des fluides, la ”Particle Image Velocimetry” ou PIV.

Figure 1: Convection d’un liquide: (a) montage expériemental, (b) manteau terrestre.

Pour réaliser cette expérience de convection, on chauffe un fil métallique par effet Joule, ce
fil étant enroulé autour d’un cylindre de plexiglas placé dans la partie inférieure de la cuve.
La mesure du courant parcourant le fil et de la tension à ses bornes permet de connâıtre, en
régime permanent, la puissance injectée dans le liquide. On caractérisera d’une part le champ
de vitesse par la méthode de PIV et d’autre part la température dans le panache thermique
grâce à un thermocouple.

1.2 Caractérisation du panache thermique

On cherche à caractériser la structure du panache thermique, c’est-à-dire la distribution de
vitesse et de température, et à estimer le flux de chaleur qui est effectivement évacué par le
mouvement de convection. On fera cette caractérisation dans deux situations : le panache
transitoire qui se forme lorsqu’on commence à alimenter en courant le fil chauffant et le
panache établi en régime permanent.

1.3 Modèle simplifié de panache thermique

Un modèle extrêmement simplifié permet d’évaluer avec peu de calculs l’ordre de grandeur
de la vitesse du fluide qui s’élève dans le panache thermique. On considère que l’écoulement
est caractérisé par la vitesse d’ascension U et par une longueur L représentative de la largeur
du panache (le gradient de vitesse est d’ordre U/L). La vitesse d’ascension peut être évaluée
en écrivant qu’elle résulte d’un équilibre entre la force de flottabilité et la friction visqueuse
ou l’inertie du fluide selon la valeur du nombre de Reynolds, Re ∼ UL/ν.

La force de flottabilité est de l’ordre de ∆ρgL3, la variation de masse volumique ∆ρ étant liée à
la variation de température : ∆ρ = ρ0α∆T , où α = 1/ρ(∂ρ/∂T ) est le coefficient d’expansion
thermique du fluide. Dans le cas de l’eau, ce coefficient est de l’ordre de 2.5 · 10−4K−1.

- Re ≪ 1, la flottabilité s’équilibre avec la force visqueuse. La contrainte (force par unité de
surface) de friction visqueuse est, en ordre de grandeur, ηU/L. La force correspondante est
donc Fv ∼ ηUL, ce qui conduit à :

U ∼ α∆TgL2

ν
(1)

- Re ≫ 1, la flottabilité s’équilibre avec la trâınée engendrée par l’inertie du fluide. La
contrainte inertielle est de l’ordre de ρU2. La trâınée s’écrit donc Fi ∼ ρU2L2, ce qui conduit
à :

U ∼ (α∆TgL)1/2 (2)

À partir de ces modèles très simples, estimez l’ordre de grandeur de U et comparez avec la
vitesse mesurée par PIV..

1.4 Un modèle plus avancé de panache thermique

Pour décrire de manière plus précise le panache thermique, il faut écrire l’équation de transport
pour la chaleur. Ici deux modes de transport de la chaleur sont mis en jeu : la diffusion
moléculaire caractérisée par le coefficient de conduction thermique λ et la convection par
l’écoulement. Le flux de chaleur dû à la diffusion est : JD = −λ∇T et le flux de chaleur
du à la convection s’écrit : JC = ρCpu(T − T0) où Cp est la chaleur spécifique par unité de
masse du fluide et T0 une température de référence. En présence de convection l’équation de
la chaleur devient :

ρCp
∂T

∂t
+∇.(JD + JC) = 0 (3)

soit, en supposant que l’écoulement obéisse toujours à la condition d’incompressibilité
∇.u = 0 (ce qui suppose que les variations locales de masse volumique soient assez petites) :

∂T

∂t
+ u.∇T =

λ

ρCp
∆T (4)

En régime stationnaire :

u.∇T = κ∆T (5)

où κ est la diffusivité thermique du fluide.
Le champ de température et le champ de vitesse sont couplés par un terme de ”flottabilité”

qui apparâıt dans l’équation de Navier-Stokes :

∂u

∂t
+ u.∇u = −1

ρ
∇p+ ν∆u+ αg(T − T0) (6)

Le panache thermique a une géométrie particulière, où la composante verticale de vitesse
u est beaucoup plus grande que la composante horizontale v et où l’échelle de longueur
horizontale δ est beaucoup plus petite que l’échelle de longueur verticale H. L’équation de
Navier Stokes, en régime stationnaire, devient donc pour la composante verticale u :

u
∂u

∂x
+ v

∂u

∂y
= −1

ρ

∂p

∂x
+ ν

∂2u

∂y2
+ αg(T − T0) (7)

et pour la composante horizontale v :

u
∂v

∂x
+ v

∂v

∂y
= −1

ρ

∂p

∂y
+ ν

∂2v

∂y2
(8)

et l’équation de la chaleur se réduit à :

u
∂T

∂x
+ v

∂T

∂y
= κ

∂2T

∂y2
(9)

Si la variation de température est petite, par conséquent la variation de masse volu-
mique est également petite, on peut considérer que le fluide reste incompressible, le terme
de variation de masse volumique n’apparaissant que dans le terme de flottabilité, moteur de
l’écoulement (approximation de Boussinesq). Le champ de vitesse obéit alors à l’équation
d’incompressibilité :

∂u

∂x
+

∂v

∂y
= 0 (10)

Si nous définissons des grandeurs caractéristiques U, V, P,Θ, X, δ, pour les composantes de
vitesse, la pression, l’écart de température et la largeur du panache, la condition d’incompressibilité
donne : U/X ∼ V/δ ce qui implique que les termes non linéaires u∂v/∂x et v∂v/∂y sont du
même ordre de grandeur. L’équation pour la composante v devient en ordre de grandeur :

UV

X
∼ P

ρδ
∼ ν

V

δ2
(11)

d’où l’ordre de grandeur de la pression : P ∼ ρU2(δ/X)2.
Dans l’eau le nombre de Prandtl Pr = ν/κ est grand devant 1 et on peut supposer que

la largeur du panache est fixée par la diffusion de la quantité de mouvement plutôt que par

la diffusion de la chaleur. On peut alors écrire, comme lorsque l’on cherche l’épaisseur d’une
couche limite classique :

Uδ2

νX
∼ 1 (12)

soit :

Uδ

ν
∼ X

δ
et

V δ

ν
∼ 1 (13)

Si on fait le même type d’analyse dimensionnelle pour l’équation sur la composante ver-
ticale de vitesse, on obtient les ordres de grandeur suivants pour les différents termes :

U2

X

U2

X

(
δ

X

)2

ν
U

δ2
αgΘ (14)

Le terme de flottabilité αgΘ est donc du même ordre de grandeur que U2/X.
Pour compléter l’analyse dimensionnelle, il faut encore examiner l’équation de transport

de la chaleur et prendre en compte le bilan global d’échange de chaleur (la chaleur injectée
au niveau de la source chaude est intégralement transportée par la panache thermique). Le
nombre de Prandtl étant grand devant 1, la diffusion de la chaleur est moins rapide que la
diffusion de la quantité de mouvement et il faut considérer une largeur δT de la distribution
de température différente de δ. Dimensionnellement, l’équation de transport de la chaleur
s’écrit donc :

UΘ

X
∼ κ

Θ

δ2T
(15)

Enfin le bilan d’énergie sur le panache s’écrit : ρCp

∫∞
−∞ u(T − T0)dx = Q où Q est la

quantité de chaleur injectée par unité de temps et par unité de longueur dans la 3è direction
(z). Soit, en ordre de grandeur : UΘδT ∼ Q/ρCp.

En combinant les différents ordres de grandeur obtenus, on peut écrire des lois d’échelle
pour la vitesse verticale dans la panache, la largeur de celui-ci et l’élévation de température
en fonction de la position verticale X :

U ∼
(

Q

ρCp
αg

)2/5

κ−1/5X1/5 (16)

δ ∼
(

Q

ρCp
αg

)−1/5

ν1/2κ1/10X2/5 et δT ∼
(

Q

ρCp
αg

)−1/5

κ3/5X2/5 (17)

Θ ∼
(

Q

ρCp

)4/5

(αg)−1/5 κ−2/5X−3/5 (18)

2 Particle Image Velocimetry (PIV)

D’une manière générale, la technique de PIV consiste à suivre le mouvement de traceurs
dispersés dans le fluide, selon une vision lagrangienne de l’écoulement. Plusieurs conditions
sont requises pour réaliser avec succès ce type de mesure:

- les particules doivent suivre fidèlement le mouvement du fluide ; on choisira donc de
petites particules (pour limiter leur inertie) d’une densité voisine de celle du liquide (pour
éviter leur sédimentation).

- la présence de traceurs ne doit pas altérer l’écoulement du fluide ; les traceurs seront
utilisés à faible concentration.

- les traceurs doivent être détectables ; on choisira des particules d’un indice optique
contrasté par rapport à celui du fluide.
Les traceurs choisis pour cette expérience sont des particules de nylon d’une taille inférieure
à 100µm. Légèrement plus denses que l’eau, elles ont tendance à sédimenter lentement. Il
faudra donc les remettre en suspension de temps en temps en agitant le liquide.

2.1 Technique de PIV

Le mouvement des traceurs est enregistré grâce à une caméra digitale. On cherche alors à
mesurer le déplacement en comparant deux images successives. La détermination du champ
de déplacement (et donc de vitesse) s’effectue en calculant la corrélation entre deux portions
d’images successives. On effectuera ce traitement dans Matlab grâce au module PIVLAB .
Le principe est le suivant: deux images successives sont découpées en portions identiques et
la fonction de corrélation entre deux portions successives est déterminée. Le décalage du pic
de corrélation par rapport au centre indique le déplacement des particules (Fig. 2).

Vδt

Vδt

Figure 2: Principe de la mesure. (a) mouvement des traceurs entre deux images, (b)
corrélation entre deux images successives

2.1.1 Acquisition des images

Pour acquérir la séquence d’images on utilise la caméra pilotée par le logiciel eBus Player
for JAI. Il faut d’abord sélectionner la caméra (GO-500M-USB) en cliquant sur le bouton
Select/Connect, puis sur le nom de la caméra dans la liste puis sur OK. Ensuite, il faut charger
la configuration de la caméra configPIV.pvcfg qui se trouve dans le dossier TP2A en cliquant
sur File puis Open.... La fréquence d’acquisition est de 40 Hz et le nombre d’images qui
seront sauvegardées est fixé à 20.

Sélectionnez le mode d’acquisition Continuous et cliquez sur Play pour visualiser en direct
le panache. Faire la mise au point sur les particules éclairées. Pour enregistrer les images,
il faut arrêter l’acquisition, aller dans la fenêtre Tools et cliquer sur Save images or Video
puis cocher Enable Image or Video Saving, définir le chemin où seront localiser les images,
privilégiez le format TIFF, cochez Save one image out of every 1 captured images. Cliquez
enfin sur OK, un bandeau rouge apparâıt alors en bas de la fenêtre d’acquisition. Il faut
ensuite changer le mode d’acquisition en sélectionnant MultiFrame. Appuyez sur Play pour
enregistrer une série de 20 images. Pour repasse en mode Live il faut décocher Enable Image
or Video Saving dans le menu Tools.

2.1.2 Analyses des images

Les images sont analysées à l’aide d’un programme Matlab (PIVlab GUI) dont le mode
opératoire est fourni dans un document que vous trouverez sur la table du TP. Il faut sauve-
garder les données dans un fichier texte (File/Save/ASCII File (x,y,u,v,vort)) qu’on relira en-
suite avec un second programme (panache2022.m) permettant d’analyser le champ de vitesse
du panache.

3 Mesure de température

Un thermocouple est placé dans la cuve. On peut le déplacer verticalement et horizontale-
ment pour estimer les profils de température à travers le panache. Cette mesure étant assez
intrusive, il faut placer le thermocouple sur le côté de la cuve lorsqu’on effectue les mesures
de vitesse par PIV.

4 Ce qu’il faut faire

A partir du champ de vitesse obtenu par PIV, mesurez l’évolution de δ et Uen fonction de la
position verticale pour plusieurs puissances. Pour chaque condition de puissance, comparez
les évolutions de la vitesse caractéristique et de la largeur du panache en fonction de X avec
celles prédites respectivement par les équations 16 et 17 à l’aide du programme panache2022.m.
Tracez ensuite l’évolution des préfacteurs des lois d’échelle en X en fonction de la puissance.
Comment interprétez-vous ces évolutions ?
A l’aide du thermocouple, réalisez un profil de température le long de la direction horizontale à
différente distances comprises entre 10 mm et 40 mm de la source de chaleur pour la puissance
maximale. Comparez ce profil à celui de la vitesse et commentez.

5 Simulation numérique

En complément de l’expérience, on se propose de faire des simulations numériques d’un
problème idéalisé. Les écoulements thermoconvectifs sont décrits par deux équations couplées
: l’équation de Navier-Stokes, où apparait un terme moteur lié à la variation spatiale de masse
volumique du fluide et l’équation de transport de la chaleur :

ρ
∂u

∂t
+ ρu.∇u = −∇p+ η∆u+ ρg

ρCp
∂T

∂t
+ ρCpu.∇T = λ∆T (19)

où la masse volumique ρ, la viscosité η, la conductivité thermique λ sont en général fonction
de la température. L’équation de conservation de la masse ne conduit plus à ∇.u = 0 puisque
la densité n’est plus uniforme.

Ce système d’équations peut être simplifié en utilisant l’approximation de Boussinesq1.
Si la variation de température est assez petite, les variations de masse volumique et des
coefficients de transport peuvent être négligés. Le seul terme du bilan de quantité de mou-
vement qui tient compte de la température est le terme de flottabilité ρg qui est à l’origine
de l’écoulement.Nous définissons une température de référence T0 à la quelle la masse volu-
mique est ρ0. La variation local de masse volumique est donnée par le coefiicient d’expansion
thermique du fluide :

α = −1

ρ

∂ρ

∂T

La masse volumique locale est : ρ = ρ0 − αρ0(T − T0) et l’équation pour la quantité de
mouvement est :

ρ0
∂u

∂t
+ ρ0u.∇u = −∇p+ η∆u+ ρ0[1− α(T − T0)]g (20)

En l’absence d’écoulement, le fluide serait en équilibre hydrostatique avec ∇pH = ρ0g, aussi
nous introduisons une pression modifiée qui est l’écart à la pression hydrostatique p′ = p−pH ,
et nous avons :

ρ0
∂u

∂t
+ ρ0u.∇u = −∇p′ + η∆u− ρ0α(T − T0)g (21)

En divisant par la masse volumique de référence, ρ0, et en utilisant l’écart à la température
de référence T ′ = T − T0, nous avons le nouveau système d’équation couplées :

∂u

∂t
+ u.∇u = −∇p′

ρ0
+ ν∆u− αT ′g

∂T ′

∂t
+ u.∇T ′ = κ∆T ′. (22)

où κ = λ/ρCp est la diffusivité thermique du fluide.

1Joseph Boussinesq (1842-1929) scientifique français qui fit des contributions majeures en hydraulique et
fut le premier à introduire la notion de viscosité turbulente

5.1 Paramètres sans dimension

Lorsque l’on développe des codes de simulation numérique, plutôt que d’utiliser les unités du
système international (kg, m, s, K), on utilise des unités pertinentes de l’écoulement que l’on
souhaite étudier. Dans le cas présent, on peut définir comme échelle de longueur L = 2R,
comme échelle de vitesse, bien que ce ne soit pas la seule possible, ν/L, comme échelle de
densité ρ0 et comme échelle de température 1/α. Les variables sans dimension sont alors
définies de la manière suivante :

u⋆ =
u

ν/L
, t⋆ =

t

L2/ν
, ∇⋆ = L∇, p⋆ =

p′

ρ0ν2/L2
, T ⋆ = αT ′. (23)

Les équations en variables adimensionnées s’écrivent alors :

ν2

L3

(
∂u⋆

∂t⋆
+ u⋆.∇⋆u⋆

)
= − ν2

L3
∇⋆p⋆ +

ν2

L3
∆⋆u⋆ + gT ⋆ey

ν

αL2

(
∂T ⋆

∂t⋆
+ u⋆.∇⋆T ⋆

)
=

κ

αL2
∆⋆T ⋆, (24)

où ey est le vecteur unitaire dirigé vers le haut. Après simplification et en supprimant le
symbole ⋆, on obtient :

∂u

∂t
+ u.∇u = −∇p+∆u+

gL3

ν2
Tey (25)

∂T

∂t
+ u.∇T =

κ

ν
∆T (26)

w

h

Figure 3: Schéma du domaine de calcul, avec identification des différentes frontières (a-e).
L’origine des coordonnées (0,0) est le point en bas à gauche du domaine de calcul.

On constate que seuls deux nombres sans dimension apparaissent dans ces équations, le
nombre de Grashof et le nombre de Prandtl :

Gr =
gL3

ν2
, P r =

ν

κ
.

Il reste à déterminer les conditions aux limites. La température étant définie comme
l’écart à T0, elle vaut 0 sur les parois verticales et horizontales du domaine de calcul (Fig.
3). La température du point chaud est une inconnue du problème, dépendant du temps et
du flux de chaleur imposé. En supposant la température uniforme dans le cylindre, le flux de
chaleur intégré sur la surface du cylindre, c’est-à-dire la puissance injectée, s’écrit :

P = −2πRHλ
∂T

∂r

∣∣∣∣
r=R

, (27)

où H est la longueur du cylindre. On obtient alors, sous forme adimensionnée :

F =
Pα

πHλ
= − ∂T

∂r

∣∣∣∣
r=1/2

(28)

La valeur de ce nombre sans dimension permet d’imposer le gradient de température à la
surface du cylindre comme condition à la limite pour l’équation de la chaleur. Les valeurs de
l’expérience étant : P ∈ [10−3, 0.03]W , H = 0.03m, λ = 0.6Wm−1K−1, α = 2.5×10−4K−1,
on obtient :

F ∈ [4.4× 10−6, 1.3× 10−4]

En outre, les nombres de Grashof et Prandtl valent :

Gr ≃ 2118960, P r = 7.

Par ailleurs, si l’on est amené, au cours de la simulation, à vouloir déterminer la valeur
d’un nombre de Reynolds pour une vitesse (numérique) Ũ , sur une longueur D̃, la viscosité
numérique ν̃ valant 1 (Cf. équation 25), on a bien :

Re =
ŨD̃

ν̃
=

U

ν/L

D

L
=

UD

ν

6 Expérience numérique

On simule ici un problème bidimensionnel, c’est à dire qu’on suppose qu’il n’y a que deux
composantes non nulles du champ de vitesse et que le problème est invariant dans la troisième
dimension de l’espace. Le schéma du domaine de calcul est représenté sur la fig. 3. Le diamètre
du point chaud est pris égal à 1. On peut faire varier la largeur et la hauteur de la cuve dans
laquelle se produit la convection. Les conditions aux limites sont les suivantes : la vitesse
du fluide est nulle sur toutes les parois solides. La température sur parois du domaine sont
égales à 0 et le flux de chaleur sur le cylindre est donné par l’équation (28).

Le logiciel utilisé pour le calcul est FreeFem++, un logiciel du domaine public développé au
laboratoire d’analyse numérique de Sorbonne Université. Les sources ainsi que les exécutables
pour Unix, Windows et MacOsX sont disponibles gratuitement. FreeFem++ permet de
résoudre des systèmes d’équations différentielles par la méthode des éléments finis en util-
isant une formulation variationnelle. Il permet également de générer automatiquement des
maillages triangulés du domaine de calcul.

6.1 Paramètres de la simulation

Il faut commencer par déterminer les paramètres sans dimension pertinents pour l’expérience
réalisée en PIV. Le point chaud a un diamètre de 6 mm, la puissance injectée est comprise
entre 10−3 et 0.03 W. Les vitesses de convection mesurées sont de quelques mm/s. Pour
l’eau le coefficient de dilatation thermique est α = 2.5 × 10−4 K−1, la viscosité cinématique
ν = 10−6 m2/s, le nombre de Prandtl est égal à 7.

6.2 Études à réaliser

On se propose ici d’étudier quelques aspects de l’instabilité thermo-convective :

Comment la géométrie d’un panache thermique évolue-t-elle avec le flux de chaleur?

Comment la géométrie d’un panache thermique rectiligne évolue-t-elle avec la distance à la
source chaude ?

Comment la stabilité du panache est-elle affectée par la proximité des parois de la cuve ?

7 Mode d’emploi détaillé

Le calcul est entièrement défini dans un fichier texte ”simul panache.edp” détaillé ci-dessous.
Ce fichier peut être modifié avec n’importe quel éditeur de texte. Sous MacOs, les éditeurs mi
et Smultron présentent l’avantage de colorer le texte en fonction de la syntaxe du programme
et d’être interfacés directement avec FreeFem.

Pour démarrer le calcul, il faut ouvrir une fenêtre terminal, se placer dans le répertoire où
se trouve le fichier simul panache.edp et exécuter la commande ”FreeFem++ simul panache.edp”.

Lorsque la commande plot est utilisée, FreeFem ouvre une autre fenêtre pour afficher le
maillage, les champs de pression, de vitesse, ...

7.1 Initialisation du calcul

• Exécuter le programme simul panache.edp

• Entrer la largeur de la cuve (20 dans l’expérience)

• Entrer la largeur de la cuve (10 dans l’expérience)

• Entrer la résolution du maillage (n = n points de maillage par unité de longueur sur les
frontières hors point chaud)

• Raffinement automatique du maillage ? Si oui, le maillage est raffiné automatiquement
à chaque pas de calcul (plus précis mais plus lent)

• Entrer le nombre de Prandtl du fluide

• Entrer la valeur du flux de chaleur

• Entrer la durée du pas de temps (en temps adimensionnel, prendre une valeur inférieure
à 1)

• Entrer le nombre d’itérations du calcul à effectuer (typiquement quelques centaines à
quelques milliers

7.2 Exécution du calcul

Une fois que le calcul est lancé, le programme affiche dans une fenêtre graphique l’évolution
temporelle du champ de température tous les dix pas de temps.

En fin de calcul, il faut appuyer sur la touche escape pour arrêter le programme.

7.2.1 Sortie des données

Pendant le calcul, une série de fichiers est écrit dans le répertoire où se trouve le fichier
simul panache.edp :

Un fichier parametres Prxx Fluxyy.txt où xx est la valeur du nombre de Prandtl et yy
la valeur du nombre de Rayleigh. Ce fichier contient les paramètres de calcul.

Tous les 10 pas de temps, fichiers vtk pour post-traitement dans Paraview.

• champs de température. Fichiers temp Fluxxx Pryy tnnn.vtk où xx est la valeur du
nombre de Rayleigh, yy, la valeur du nombre de Prandtl et nnn est le numéro d’itération

• champs de vitesse. Fichiers vit Fluxxx Pryy tnnn.vtk

• champs de flux de chaleur. Fichiers flux Fluxxx Pryy tnnn.vtk. Le flux de chaleur
est calculé par : j = uT −∇T en tenant compte des variables adimensionnelles.

Un fichier flux Fluxxx Pryy.txt qui contient l’évolution au cours du temps du flux de
chaleur intégré sur la surface du point chaud.

7.3 Post traitement

Les fichiers .vtk peuvent être lus dans le logiciel Paraview.

8 Programme utilisé pour le calcul

// panache thermique

//adapte du programme boussinesq.edp de Olivier Pironneau

load "iovtk" ;

string yes;

// definition de la geometrie.

// Le diametre du cylindre chaud est pris comme unite de longueur

real he=1, wi=1;

cout << " Entrer la largeur de la cuve (rapport largeur/diametre du cylindre) :"; cin >> wi;

cout << " Entrer la hauteur de la cuve (rapport hauteur/diametre du cylindre) :"; cin >> he;

// definition des limites du domaine de calcul

border a(t=0,wi) {x=t;y=0;}; // fond

border b(t=0,he) {x=wi;y=t;}; // cote droit

border c(t=0,wi) {x=wi-t;y=he;}; // haut

border d(t=0,he) {x=0;y=he-t;}; // cote gauche

// cylindre (parcouru dans le sens trigo inverse)

// le centre du cylindre est a 1,5 diametre au-dessus du fond

border e(t=2*pi,0) {x=0.5*wi+0.5*cos(t);y=1.5+0.5*sin(t);label=5;};

int nm=1;

cout << " Entrer la resolution du maillage (>=1) :"; cin >> nm;

int nm1,nm2,nm3;

nm1=floor(wi*nm);

nm2=floor(he*nm);

mesh Th=buildmesh(a(nm1)+b(nm2)+c(nm1)+d(nm2)+e(12*nm));

plot(Th,wait=1);

// raffinement automatique du maillage ou pas ?

bool refinemesh=false;

cout << " Raffinement automatique du maillage pendant le calcul ? (o,n)"; cin >> yes;

if (yes == "o") refinemesh=true;

// espaces d’elements finis sur le maillage Th :

// P2 pour la vitesse,

// P1 pour la pression et la temperature

fespace Vh(Th,P2), Qh(Th,P1);

Vh u=0,v=0,uu,vv,up,vp; // champ de vitesse

Vh jx,jy ; // flux de chaleur

Qh p=0,pp; // champ de pression

Qh r=0,rr,rp; // champ de temperature

Qh gtx,gty; // gradient de temperature

real fc; // flux de chaleur sur le point chaud

int n=0;

// definition des parametres physiques

real Pr ;

cout << " Entrer le nombre de Prandtl (7 pour l’eau):"; cin >> Pr;

real F;

cout << " Entrer le flux de chaleur adimensionne:"; cin >> F;

//real Re ;

//cout << " Entrer le nombre de Reynolds :"; cin >> Re;

//real Pe=Re*Pr;

//cout << " Nombre de Peclet :" << Pe << "\n";

//real rayleigh ;

//cout << " Entrer le nombre de Rayleigh :"; cin >> rayleigh;

//real g=rayleigh/Pr; // nombre de Grashof

real GR;

GR = 2118960;

cout << " nombre de Grashof :" << GR << "\n";

// gravite inverse ?

//cout << " Inversion de la gravite ? (o,n)"; cin >> yes;

//if (yes == "o") g=-g;

real dt;

cout << " Entrer la duree du pas de temps :"; cin >> dt;

//dt = 0.1;

int nit ;

cout << " Entrer le nombre d iterations :"; cin >> nit;

//nit = 1000;

// definition des valeurs pour la visualisation de la temperature

real[int] tempval(51);

real vmax=1;

int i,j;

for (i=0;i<51;i++){

tempval[i]=vmax*0.02*i;

}

// definition des couleurs pour l’affichage du champ de temperature

real[int] colors(150);

for (i=0;i<50;i++){

colors[3*i]=0.75-0.02*i ;

colors[3*i+1]= 1;

colors[3*i+2]= 1 ;

}

// definition des tableaux pour les profils de vitesse et de temperature

real[int] pos(100),profvitv(100),proftempv(100),profvith(100),proftemph(100);

for(i=0;i<100;i++){

pos[i]=0.01*i;

}

// definition des fichiers pour la sortie des profils

string profhvfile="profilhv_Flux"+F+".txt" ;

ofstream profhv(profhvfile,append);

string profhtfile="profilht_Flux"+F+".txt" ;

ofstream profht(profhtfile,append);

// fichier pour l’evolution du flux de chaleur sur le point chaud

string fluxfile="flux_Flux"+F+"_Pr"+Pr+".txt";

ofstream flux(fluxfile,append);

// fichiers de sortie des donnees au format vtk pour affichage avec Paraview

string vtkfile,temptkfile,fluxvtk;

//real T=0.5, nudt=dt/Re, peclet=Pr*Re, pedt=dt/peclet;

real prdt=dt/Pr;

// sauvegarde des parametres de calcul dans un fichier

string paramfile="parametres_Pr"+Pr+"_flux"+F+".txt" ;

ofstream param(paramfile,append);

param << "largeur " << ":" << wi << "\n";

param << "hauteur " << ":" << he << "\n";

param << "prandtl " << ":" << Pr << "\n";

param << "Flux" << ":" << F << "\n";

//param << "rayleigh" << "," << rayleigh << "\n";

param << "pas de temps " << ":" << dt << ", nb iterations" << ":" << nit <<"\n";

// iteration en temps

for (n=0;n<nit;n+=1){

up=u;vp=v; rp=r;

solve NS([u,v,p],[uu,vv,pp],solver=Crout,init=n) =

int2d(Th) (u*uu+v*vv

+ dt*(dx(u)*dx(uu)+dy(u)*dy(uu)

+ dx(v)*dx(vv)+dy(v)*dy(vv))

+ p*pp*1.e-6

- dt*(p*(dx(uu)+dy(vv))+pp*(dx(u)+dy(v))))

- int2d(Th)(convect([up,vp],-dt,up)*uu

+ convect([up,vp],-dt,vp)*vv

+ r*GR*vv)

+ on(a,b,c,d,e,u=0,v=0); // conditions aux limites : vitesse nulle sur les parois solides

solve thermal(r,rr,solver=Crout,init=n)=

int2d(Th) (r*rr+prdt*(dx(r)*dx(rr)+dy(r)*dy(rr)))

- int2d(Th)(convect([up,vp],-dt,rp)*rr)

+ int1d(Th,5) (- F * rr)

+ on(c,r=0);

//+ on(c,r=0) + on(e,r=1); // conditions aux limites pour la temperature : r=1 point chaud, r=0 haut de la boite

// profils horizontaux

for(i=0;i<100;i++){

y=0.5*he;

x=wi*(0.5+0.01*(i-50));

profvith[i]=v;

proftemph[i]=r;

}

if ( !(n % 1) && (n>0)) {

vtkfile="vit_Flux"+F+"_Pr"+Pr+"_t"+n+".vtk";

savevtk(vtkfile,Th,[u,v],dataname="vitesse");

//plot(r,fill=true,hsv=colors,cmm="t="+n,ps="temp_re_"+Re+"t="+n+".eps");

plot(r,fill=true,hsv=colors,cmm="t="+n);

temptkfile="temp_Flux"+F+"_Pr"+Pr+"_t"+n+".vtk";

savevtk(temptkfile,Th,r,dataname="temperature");

//plot(r,fill=true,hsv=colors,cmm="t="+n);

// calcul du flux de chaleur

gtx=dx(r);

gty=dy(r);

jx=u*r-gtx;

jy=v*r-gty;

fluxvtk="flux_Flux"+F+"_Pr"+Pr+"_t"+n+".vtk";

savevtk(fluxvtk,Th,[jx,jy],dataname="flux");

// calcul du flux de chaleur a la surface du point chaud

fc=-int1d(Th,5) (gtx*N.x+gty*N.y);

flux << fc << "\n";

for (j=0;j<100;j++){

profhv << profvith[j] << ",";

}

profhv << "\n" ;

for (j=0;j<100;j++){

profht << proftemph[j] << ",";

}

profht << "\n" ;

}

if (refinemesh) {

// adaptation du maillage et interpolation des fonctions sur le nouveau maillage

Th=adaptmesh(Th,r,hmin=0.05); // taille minimale des mailles = 0.05

u=u;

v=v;

p=p;

r=r;

}

}

9 Formulation variationnelle

Pour décrire ce qu’est la formulation variationnelle d’une équation différentielle [?], con-
sidérons le problème d’une poutre flexible, posée à ses extrémités (mais libre en rotation), de
longueur unité soumise à une charge répartie f . Le moment fléchissant local u(x) est décrit
par l’équation différentielle de Poisson qui s’écrit, une fois adimensionnée :

−u”(x) = f(x) (29)

avec les conditions aux limites u(0) = u(1) = 0. En multipliant cette équation par une
fonction continue et différentiable v(x), on obtient :

−
∫ 1

0
u”(x)v(x)dx =

∫ 1

0
f(x)v(x) (30)

En intégrant par parties le membre de gauche, on a :∫ 1

0
u′(x)v′(x)dx− [u”v]10 =

∫ 1

0
f(x)v(x) (31)

Si v est choisi telle que v(0) = v(1) = 0 :∫ 1

0
u′(x)v′(x)dx =

∫ 1

0
f(x)v(x) (32)

Le problème variationnel consiste à trouver la fonction u qui satisfait l’équation 32 quelle que
soit la fonction v.

La méthode des éléments finis consiste à trouver une solution sous la forme : u(x) =∑N
i=1 uiϕi(x) où les fonctions ϕi sont N fonctions linéairement indépendantes et ui sont N

nombres réels à déterminer. En prenant v = ϕj , le problème 32 se ramène à trouver les
nombres ui tels que :

N∑
i=1

ui

(∫ 1

0
ϕ′
i(x)ϕ

′
j(x)dx

)
=

∫ 1

0
f(x)ϕ′

j(x)dx (33)

pour tout j compris entre 1 et N . On définit la matrice de rigidité N × N A telle que
Aji =

∫ 1
0 ϕ′

i(x)ϕ
′
j(x)dx. Si u est le vecteur de composantes ui et f le vecteur de composantes

fj =
∫ 1
0 f(x)ϕ′

j(x)dx, le problème se ramène à résoudre le système linéaire :

Au = f (34)

et donc à inverser la matrice A.
Dans la méthodes des éléments finis de degré 1, on divise l’intervalle [0, 1] en N+1 parties

de largeur h = 1/(N + 1) et commençant respectivement en xi = ih. On définit les fonctions
ϕi(x) ”triangulaires”telles que :

ϕi(x) =


x−xi−1

h si xi−1 < x < xi
x−xi+1

h si xi < x < xi+1

0 si x <= xi−1 ou x >= xi+1

(35)

Ceci revient à faire une approximation de la solution par un ensemble de N fonctions affines
par morceaux.

De la même manière, on peut définir des éléments de degré 2 où les fonctions ϕi sont des
fonctions quadratiques de x, la solution approchée étant une succession d’arcs de parabole.
Ces résultats à une dimension peuvent être étendus aux cas bidimensionnel et tridimensionnel.

